Linear stability of long waves in two-layer channel flow

This notebook has been written in *Mathematica* by

Mark J. McCready
Professor and Chair of Chemical Engineering
University of Notre Dame
Notre Dame IN 46556
USA

Mark.J.McCready.1@nd.edu
http://www.nd.edu/~mjm/

Version: 6/13/98

This notebook solves the longwave stability problem for two-layer, pressure driven channel flow

References:
C. -S Yih (1967) "Instability due to viscosity stratification", J. Fluid Mech., **27** pp337-352**.**
S. G. Yiantsios and B. G. Higgins (1988) "Linear stability of plane Poiseullie flow of two-superposed fluids", Phys. Fluids**, 31** pp3225-3238**.**

The 0 order and first order terms for the wave velocity for a pressure driven channel flow are obtained from a long wave expansion. All of the necessary manipulations are shown for this direct perturbation solution method. I plan to add the solution method using the adjoint system (see P. J. Blennerhassett, Trans. Roy. Soc. 298, pp451-494 (1980)) sometime in the next few months.

■ The base state flow

This notebook addresses the case of Poiseulle flow so that the walls are fixed.
The upper fluid is 1 and the lower is 2, the velocity at the interface, y=0, is used to normalize the velocity so that we have

```
u1 = 1 + a1 y + b1 y^2;

u2 = 1 + a2 y + b2 y^2;
```

```
The constants are given here but we will substitute these as needed later
```

$$a1 = \frac{m - n^2}{n^2 + n};$$

$$b1 = -\frac{m + n}{n^2 + n};$$

$$a2 = \frac{a1}{m};$$

$$b2 = \frac{b1}{m};$$

The m's and n's are the viscosity ratio, m= $\frac{\mu_2}{\mu_1}$ and the thickness ratio, n= $\frac{d_2}{d_1}$.

```
profile1 = Plot[u1 //. {a1 -> (m - n^2) / (n^2 + n),
    a2 -> a1 / m, b1 -> -(m + n) / (n^2 + n),
    b2 -> b1 / m, m → 20, n → 1/2}, {y, 0, 1}, PlotRange → All,
  DisplayFunction -> Identity]
```

- Graphics -

```
profile2 = Plot[u2 //. {a1 -> (m - n^2) / (n^2 + n),
    a2 -> a1 / m, b1 -> -(m + n) / (n^2 + n),
    b2 -> b1 / m, m → 20, n → 1/2}, {y, -1/2, 0}, PlotRange → All]

vprofile = Show[profile1, profile2,
   DisplayFunction -> $DisplayFunction]
```



- Graphics -

This looks OK for the case of the lower fluid (phase 2) being more viscous.

■ The stability equations for the disturbance flow

The governing equations will be the Orr-Sommerfeld equation for each phase.

The equations for the disturbance stream functions in both phases are the Orr-Sommerfeld equation.

Upper phase

$$\mathbf{os1} = \phi^{(4)}[y] - 2\,\alpha^2\,\phi''[y] + \alpha^4\,\phi[y] - I\,\alpha\,R\,((U_1[y] - c)\,(\phi''[y] - \alpha^2\,\phi[y]) - U_1''[y]\,\phi[y])$$

$$\phi(y)\,\alpha^4 - 2\,\phi''(y)\,\alpha^2 - i\,R\,((U_1[y] - c)\,(\phi''(y) - \alpha^2\,\phi(y)) - \phi(y)\,U_1''(y))\,\alpha + \phi^{(4)}(y)$$

lower phase

$$\mathbf{os2} = \psi^{(4)}[y] - 2\,\alpha^2\,\psi''[y] + \alpha^4\,\psi[y] - \frac{1}{m}\,(I\,\alpha\,R\,r\,((U_2[y] - c)\,(\psi''[y] - \alpha^2\,\psi[y]) - U_2''[y]\,\psi[y]))$$

$$\psi(y)\,\alpha^4 - 2\,\psi''(y)\,\alpha^2 - \frac{i\,r\,R\,((U_2[y] - c)\,(\psi''(y) - \alpha^2\,\psi(y)) - \psi(y)\,U_2''(y))\,\alpha}{m} + \psi^{(4)}(y)$$

Some narration about boundary conditions

Here is a brief discussion of the boundary conditions.

First the velocity is 0 on the walls
*** $\phi[0] = \phi'[0] = 0$,
*** $\psi[-n] = \psi'[-n] = 0$

Next the tangential velocity at the interface

Note that this says that u1(η) = u2(η). Thus we must expand the velocity to include the effect of the average velocity

*** $\phi'[0] + \eta\, \overline{U1}'[y] = \psi'[0] + \eta\, \overline{U2}'[y]$

this is

η is given from the Kinematic condition D[η,t] = $v_y[0]$

substituting gives:

*** $\phi'[0]-\psi'[0] = \phi[0](a2-a1)/(c-\overline{U}[0])$

Next the normal velocity at the interface

d[ψ,x] = d[ϕ,x] which gives simply

*** $\psi[0] = \phi[0]$

Next, the tangential stress is continuous across the interface

$\mu_1 (\partial u1/\partial y + \partial v1/\partial x) = \mu_2 (\partial u2/\partial y + \partial v2/\partial x)$

which gives

*** $\phi''[0] + \alpha^2\, \phi[0] = m\, (\psi''[0] + \alpha^2\, \psi[0])$

Finally, we have the continuity of Normal stress. This is comprised of, in dimensional terms

$(p - \rho\, g\, a - \mu\, dv/dy\,) = S\, \partial^2 \eta/\partial x^2$

which says that the pressure change differs by the surface tension jump.

We substitute for the pressure from the governing equation and make some other simplifications and get

$$-i\, \alpha\, R\ ((c-\overline{U}[0])\, \phi'[0] + a1\, \phi[0]) - \phi'''[0] + 3\, \alpha^2\, \phi'[0] +$$
$$i\, \alpha\, R\, r\, ((c-\overline{U}[0])\, \psi'[0] + a2\, \psi[0]) + m(\psi'''[0] - 2\, \alpha^2\, \psi'[0]) =$$
$$I\, \alpha\, R\, (F + \alpha^2\, S)\, \phi[0]/(c-\overline{U}[0])$$

■ Expansions for the stream functions and wave speed

The problem will now be solved using the long wave expansion, i.e. α-->0 so that that we can write the stream functions and the wave speed as perturbation expansions in α.

```
ϕexp = ϕ0[y] + α ϕ1[y] + α^2 ϕ2[y];
ψexp = ψ0[y] + α ψ1[y] + α^2 ψ2[y];
cexp = c0 + α c1 + α^2 c2;
```

— *General::spell1 : Possible spelling error: new symbol name "ψexp" is similar to existing symbol "ϕexp".*

The procedure is to expand the governing equations and the boundary conditions in the expansion variables. These can then be collected in powers of α. Each separate power of α must be independent which allows a sequential solution to the problem. Higher orders depend on known solutions to lower orders.

■ Expand the upper phase os-equ

```
os11 = os1 /. {ϕ^(a1_)[y] :→ ∂{y,a1} ϕexp, c → cexp, ϕ[y] → ϕexp,
   U1″[y] → D[u1, {y, 2}], U1[y] -> u1}
```

$$(\phi_2[y]\,\alpha^2 + \phi_1[y]\,\alpha + \phi_0[y])\,\alpha^4 - 2\,(\phi_2''(y)\,\alpha^2 + \phi_1''(y)\,\alpha + \phi_0''(y))\,\alpha^2 + \phi_2^{(4)}(y)\,\alpha^2 - i\,R\,((\text{b1}\,y^2 + \text{a1}\,y - \text{c2}\,\alpha^2 - \text{c0} - \text{c1}\,\alpha + 1)\,(-(\phi_2[y]\,\alpha^2 + \phi_1[y]\,\alpha + \phi_0[y])\,\alpha^2 + \phi_2''(y)\,\alpha^2 + \phi_1''(y)\,\alpha + \phi_0''(y)) - 2\,\text{b1}\,(\phi_2[y]\,\alpha^2 + \phi_1[y]\,\alpha + \phi_0[y]))\,\alpha + \phi_1^{(4)}(y)\,\alpha + \phi_0^{(4)}(y)$$

```
os12 = Expand[os11];
```

— *General::spell1 : Possible spelling error: new symbol name "os12" is similar to existing symbol "os2".*

```
os13 = Collect[os12, α]
os1O0 = Coefficient[os13, α, 0]
```

$$\phi_0^{(4)}(y)$$

```
os1O1 = Simplify[Coefficient[os13, α, 1]]
```

— *General::spell1 : Possible spelling error: new symbol name "os1O1" is similar to existing symbol "os11".*

$$-i\,(-2\,\text{b1}\,R\,\phi_0[y] + R\,(\text{b1}\,y^2 + \text{a1}\,y - \text{c0} + 1)\,\phi_0''(y) + i\,\phi_1^{(4)}(y))$$

```
os1O2 = Simplify[Coefficient[os13, α, 2]]
```

— *General::spell1 : Possible spelling error: new symbol name "os1O2" is similar to existing symbol "os12".*

$$-i\,(\text{b1}\,R\,\phi_1''(y)\,y^2 + \text{a1}\,R\,\phi_1''(y)\,y - 2\,\text{b1}\,R\,\phi_1[y] - (\text{c1}\,R + 2\,i)\,\phi_0''(y) - \text{c0}\,R\,\phi_1''(y) + R\,\phi_1''(y) + i\,\phi_2^{(4)}(y))$$

- **Expand the lower phase os-equ**

```
os21 = os2 /. {ψ^(a1_)[y] :→ ∂_{y,a1} ψexp, c → cexp, ψ[y] → ψexp,
    U2''[y] → D[u2, {y, 2}], U2[y] -> u2}
```

— *General::spell : Possible spelling error: new symbol name "os21" is similar to existing symbols {os1, os12}.*

$$(\psi_2[y]\,\alpha^2 + \psi_1[y]\,\alpha + \psi_0[y])\,\alpha^4 - 2\,(\psi_2''(y)\,\alpha^2 + \psi_1''(y)\,\alpha + \psi_0''(y))\,\alpha^2 + \psi_2^{(4)}(y)\,\alpha^2 - \frac{1}{m}\,(i\,r\,R\,((\mathrm{b2}\,y^2 + \mathrm{a2}\,y - \mathrm{c2}\,\alpha^2 - \mathrm{c0} - \mathrm{c1}\,\alpha + 1)\,(-(\psi_2[y]\,\alpha^2 + \psi_1[y]\,\alpha + \psi_0[y])\,\alpha^2 + \psi_2''(y)\,\alpha^2 + \psi_1''(y)\,\alpha + \psi_0''(y)) - 2\,\mathrm{b2}\,(\psi_2[y]\,\alpha^2 + \psi_1[y]\,\alpha + \psi_0[y]))\,\alpha) + \psi_1^{(4)}(y)\,\alpha + \psi_0^{(4)}(y)$$

```
os22 = Expand[os21];
```

```
os23 = Collect[os22, α];
```

```
os2O0 = Coefficient[os23, α, 0]
```

$$\psi_0^{(4)}(y)$$

```
os2O1 = Simplify[Coefficient[os23, α, 1]]
```

— *General::spell1 : Possible spelling error: new symbol name "os2O1" is similar to existing symbol "os21".*

$$-\frac{i\,(-2\,\mathrm{b2}\,r\,R\,\psi_0[y] + r\,R\,(\mathrm{b2}\,y^2 + \mathrm{a2}\,y - \mathrm{c0} + 1)\,\psi_0''(y) + i\,m\,\psi_1^{(4)}(y))}{m}$$

```
os2O2 = Coefficient[os23, α, 2]
```

— *General::spell1 : Possible spelling error: new symbol name "os2O2" is similar to existing symbol "os22".*

$$-\frac{i\,\mathrm{b2}\,r\,R\,\psi_1''(y)\,y^2}{m} - \frac{i\,\mathrm{a2}\,r\,R\,\psi_1''(y)\,y}{m} + \frac{2\,i\,\mathrm{b2}\,r\,R\,\psi_1[y]}{m} + \frac{i\,\mathrm{c1}\,r\,R\,\psi_0''(y)}{m} - 2\,\psi_0''(y) + \frac{i\,\mathrm{c0}\,r\,R\,\psi_1''(y)}{m} - \frac{i\,r\,R\,\psi_1''(y)}{m} + \psi_2^{(4)}(y)$$

- **expand the boundary conditions**

Each of these boundary conditions are written so that the expression is equal to 0.

Here is no tang velocity at top wall, bc1

```
bc1 = ϕ'[y]
```

$$\phi'(y)$$

```
bc11 = bc1 /. {ϕ^(a1_)[y] :> ∂{y,a1} ϕexp, ψ^(a1_)[y] :> ∂{y,a1} ψexp,
    c → cexp, ϕ[y] → ϕexp, U1''[y] → 2 b1, U0 → 1}
```

$\phi_2'(y)\,\alpha^2 + \phi_1'(y)\,\alpha + \phi_0'(y)$

```
bc12 = bc11 /. y → 1
```

$\phi_2'(1)\,\alpha^2 + \phi_1'(1)\,\alpha + \phi_0'(1)$

```
bc13 = Expand[bc12]
```

$\phi_2'(1)\,\alpha^2 + \phi_1'(1)\,\alpha + \phi_0'(1)$

```
bc1O0 = Coefficient[bc13, α, 0]
```

$\phi_0'(1)$

```
bc1O1 = Coefficient[bc13, α, 1]
```

— *General::spell1 : Possible spelling error: new symbol name "bc1O1" is similar to existing symbol "bc11".*

$\phi_1'(1)$

```
bc1O2 = Coefficient[bc13, α, 2]
```

— *General::spell1 : Possible spelling error: new symbol name "bc1O2" is similar to existing symbol "bc12".*

$\phi_2'(1)$

Here is no norm velocity at top wall, bc2

```
bc2 = ϕ[y]
```

$\phi(y)$

```
bc21 = bc2 /. {ϕ^(a1_)[y] :> ∂{y,a1} ϕexp, ψ^(a1_)[y] :> ∂{y,a1} ψexp,
    c → cexp, ϕ[y] → ϕexp, U1''[y] → 2 b1, U0 → 1}
```

— *General::spell : Possible spelling error: new symbol name "bc21" is similar to existing symbols {bc1, bc12}.*

$\phi_2[y]\,\alpha^2 + \phi_1[y]\,\alpha + \phi_0[y]$

```
bc22 = bc21 /. y → 1
```

$\phi_2[1]\,\alpha^2 + \phi_1[1]\,\alpha + \phi_0[1]$

```
bc23 = Expand[bc22]
```

$\phi_2[1]\,\alpha^2 + \phi_1[1]\,\alpha + \phi_0[1]$

```
bc2O0 = Coefficient [bc23, α, 0]
```

$\phi_0[1]$

```
bc2O1 = Coefficient [bc23, α, 1]
```

— *General::spell1 : Possible spelling error: new symbol name "bc2O1" is similar to existing symbol "bc21".*

$\phi_1[1]$

```
bc2O2 = Coefficient [bc23, α, 2]
```

— *General::spell1 : Possible spelling error: new symbol name "bc2O2" is similar to existing symbol "bc22".*

$\phi_2[1]$

Here is the tangential velocity match, bc3

$$\text{bc3} = \phi'[y] - \psi'[y] - \frac{\phi[y]\,(\text{a2} - \text{a1})}{c - U_0}$$

$$-\frac{(\text{a2} - \text{a1})\,\phi(y)}{c - U_0} + \phi'(y) - \psi'(y)$$

$$\text{bc31} = \text{bc3} \;/.\; \{\phi^{(\text{a1}\_)}[y] :\to \partial_{\{y,\text{a1}\}}\phi\text{exp},\; \psi^{(\text{a1}\_)}[y] :\to \partial_{\{y,\text{a1}\}}\psi\text{exp},\; c \to \text{cexp},\; \phi[y] \to \phi\text{exp},\; U_1''[y] \to 2\,\text{b1},\; U_0 \to 1\}$$

— *General::spell : Possible spelling error: new symbol name "bc31" is similar to existing symbols {bc1, bc13}.*

$$\phi_2'(y)\,\alpha^2 - \psi_2'(y)\,\alpha^2 + \phi_1'(y)\,\alpha - \psi_1'(y)\,\alpha - \frac{(\text{a2} - \text{a1})\,(\phi_2[y]\,\alpha^2 + \phi_1[y]\,\alpha + \phi_0[y])}{\text{c2}\,\alpha^2 + \text{c1}\,\alpha + \text{c0} - 1} + \phi_0'(y) - \psi_0'(y)$$

```
bc32 = bc31 /. y → 0
```

— *General::spell : Possible spelling error: new symbol name "bc32" is similar to existing symbols {bc2, bc23}.*

$$\phi_2'(0)\,\alpha^2 - \psi_2'(0)\,\alpha^2 + \phi_1'(0)\,\alpha - \psi_1'(0)\,\alpha - \frac{(\text{a2} - \text{a1})\,(\phi_2[0]\,\alpha^2 + \phi_1[0]\,\alpha + \phi_0[0])}{\text{c2}\,\alpha^2 + \text{c1}\,\alpha + \text{c0} - 1} + \phi_0'(0) - \psi_0'(0)$$

```
bc33 = Expand [bc32] ;
bc34 = Simplify [Series [bc33, {α, 0, 3}]]
bc3O0 = Coefficient [bc34, α, 0]
```

$$\frac{\text{a2}\,\phi_0[0]}{1 - \text{c0}} + \frac{\text{a1}\,\phi_0[0]}{\text{c0} - 1} + \phi_0'(0) - \psi_0'(0)$$

```
bc3O1 = Coefficient [bc34, α, 1]
```

— *General::spell1 : Possible spelling error: new symbol name "bc3O1" is similar to existing symbol "bc31".*

$$-\frac{\text{a1}\,(\text{c1}\,\phi_0[0] - (\text{c0} - 1)\,\phi_1[0])}{(\text{c0} - 1)^2} + \frac{\text{a2}\,(\text{c1}\,\phi_0[0] - (\text{c0} - 1)\,\phi_1[0])}{(\text{c0} - 1)^2} + \phi_1'(0) - \psi_1'(0)$$

```
bc3O2 = Coefficient[bc34, α, 2]
```

— *General::spell1 : Possible spelling error: new symbol name "bc3O2" is similar to existing symbol "bc32".*

$$\frac{1}{(c0-1)^3}\,((\phi_2'(0)-\psi_2'(0))\,(c0-1)^3 + a1\,(\phi_0[0]\,c1^2-(c0-1)\,\phi_1[0]\,c1-(c0-1)\,(c2\,\phi_0[0]-c0\,\phi_2[0]+\phi_2[0])) - a2\,(\phi_0[0]\,c1^2-(c0-1)\,\phi_1[0]\,c1-(c0-1)\,(c2\,\phi_0[0]-c0\,\phi_2[0]+\phi_2[0])))$$

Here is the normal velocity match,bc4

```
bc4 = ψ[y] - ϕ[y]
```

$$\psi(y)-\phi(y)$$

```
bc41 = bc4 /. {ϕ^(a1_)[y] :→ ∂_{y,a1} ϕexp, ψ^(a1_)[y] :→ ∂_{y,a1} ψexp,
    c → cexp, ϕ[y] → ϕexp, ψ[y] → ψexp, U_1''[y] → 2 b1, U_0 → 1}
```

— *General::spell1 : Possible spelling error: new symbol name "bc41" is similar to existing symbol "bc1".*

$$-\phi_2[y]\,\alpha^2+\psi_2[y]\,\alpha^2-\phi_1[y]\,\alpha+\psi_1[y]\,\alpha-\phi_0[y]+\psi_0[y]$$

```
bc42 = bc41 /. y → 0
```

— *General::spell1 : Possible spelling error: new symbol name "bc42" is similar to existing symbol "bc2".*

$$-\phi_2[0]\,\alpha^2+\psi_2[0]\,\alpha^2-\phi_1[0]\,\alpha+\psi_1[0]\,\alpha-\phi_0[0]+\psi_0[0]$$

```
bc43 = Expand[bc42];
```

— *General::spell : Possible spelling error: new symbol name "bc43" is similar to existing symbols {bc3, bc34}.*

```
bc4O0 = Coefficient[bc43, α, 0]
```

$$\psi_0[0]-\phi_0[0]$$

```
bc4O1 = Coefficient[bc43, α, 1]
```

— *General::spell1 : Possible spelling error: new symbol name "bc4O1" is similar to existing symbol "bc41".*

$$\psi_1[0]-\phi_1[0]$$

```
bc4O2 = Coefficient[bc43, α, 2]
```

— *General::spell1 : Possible spelling error: new symbol name "bc4O2" is similar to existing symbol "bc42".*

$$\psi_2[0]-\phi_2[0]$$

Here is the interfacial shear stress match,bc5

```
bc5 = ϕ''[y] + α^2 ϕ[y] - m (ψ''[y] + α^2 ψ[y])
```

$$\phi(y)\,\alpha^2+\phi''(y)-m\,(\psi(y)\,\alpha^2+\psi''(y))$$

```
bc51 = bc5 /. {ϕ^(a1_)[y] :→ ∂_{y,a1} ϕexp, ψ^(a1_)[y] :→ ∂_{y,a1} ψexp,
    c → cexp, ϕ[y] → ϕexp, ψ[y] → ψexp, U1″[y] → 2 b1, U0 → 1}
```

— *General::spell1 : Possible spelling error: new symbol name "bc51" is similar to existing symbol "bc1".*

$$(\phi_2[y]\,\alpha^2+\phi_1[y]\,\alpha+\phi_0[y])\,\alpha^2+\phi_2''(y)\,\alpha^2+\phi_1''(y)\,\alpha+\phi_0''(y)-m\,((\psi_2[y]\,\alpha^2+\psi_1[y]\,\alpha+\psi_0[y])\,\alpha^2+\psi_2''(y)\,\alpha^2+\psi_1''(y)\,\alpha+\psi_0''(y))$$

```
bc52 = bc51 /. y → 0
```

— *General::spell1 : Possible spelling error: new symbol name "bc52" is similar to existing symbol "bc2".*

$$(\phi_2[0]\,\alpha^2+\phi_1[0]\,\alpha+\phi_0[0])\,\alpha^2+\phi_2''(0)\,\alpha^2+\phi_1''(0)\,\alpha+\phi_0''(0)-m\,((\psi_2[0]\,\alpha^2+\psi_1[0]\,\alpha+\psi_0[0])\,\alpha^2+\psi_2''(0)\,\alpha^2+\psi_1''(0)\,\alpha+\psi_0''(0))$$

```
bc53 = Expand[bc52];
```

— *General::spell1 : Possible spelling error: new symbol name "bc53" is similar to existing symbol "bc3".*

```
bc54 = Collect[bc53, α];
```

— *General::spell1 : Possible spelling error: new symbol name "bc54" is similar to existing symbol "bc4".*

```
bc5O0 = Coefficient[bc54, α, 0]
```

$$\phi_0''(0)-m\,\psi_0''(0)$$

```
bc5O1 = Coefficient[bc54, α, 1]
```

— *General::spell1 : Possible spelling error: new symbol name "bc5O1" is similar to existing symbol "bc51".*

$$\phi_1''(0)-m\,\psi_1''(0)$$

```
bc5O2 = Coefficient[bc54, α, 2]
```

— *General::spell1 : Possible spelling error: new symbol name "bc5O2" is similar to existing symbol "bc52".*

$$\phi_0[0]-m\,\psi_0[0]+\phi_2''(0)-m\,\psi_2''(0)$$

Here is the pressure boundary condition,bc6

```
bc6 = -I α R ((c - U0) ϕ′[y] + a1 ϕ[y]) - ϕ^(3)[y] + 3 α^2 ϕ′[y] +
   I α R r ((c - U0) ψ′[y] + a2 ψ[y]) + m (ψ^(3)[y] - 3 α^2 ψ′[y]) -
   (I α R (F + α^2 S) ϕ[y]) / (c - U0)
```

$$3\,\phi'(y)\,\alpha^2-\frac{i\,R\,(S\,\alpha^2+F)\,\phi(y)\,\alpha}{c-U_0}-i\,R\,(\mathrm{a1}\,\phi(y)+(c-U_0)\,\phi'(y))\,\alpha+i\,r\,R\,(\mathrm{a2}\,\psi(y)+(c-U_0)\,\psi'(y))\,\alpha-\phi^{(3)}(y)+m\,(\psi^{(3)}(y)-3\,\alpha^2\,\psi'(y))$$

```
bc61 = bc6 /. {ϕ^(a1_)[y] :→ ∂_{y,a1} ϕexp, ψ^(a1_)[y] :→ ∂_{y,a1} ψexp,
    c → cexp, ϕ[y] → ϕexp, ψ[y] → ψexp, U1″[y] → 2 b1, U0 → 1}
bc62 = bc61 /. y → 0
```

```
bc63 = Expand[bc62];
```

— *General::spell1 : Possible spelling error: new symbol name "bc63" is similar to existing symbol "bc3".*

```
bc64 = Series[bc63, {α, 0, 3}]
bc6O0 = Coefficient[bc64, α, 0]
```

$m\,\psi_0^{(3)}(0) - \phi_0^{(3)}(0)$

```
bc6O1 = Coefficient[bc64, α, 1]
```

— *General::spell1 : Possible spelling error: new symbol name "bc6O1" is similar to existing symbol "bc61".*

$$-i\,\mathrm{a1}\,R\,\phi_0[0] - \frac{i\,F\,R\,\phi_0[0]}{\mathrm{c0}-1} + i\,\mathrm{a2}\,r\,R\,\psi_0[0] - i\,\mathrm{c0}\,R\,\phi_0'(0) + i\,R\,\phi_0'(0) + i\,\mathrm{c0}\,r\,R\,\psi_0'(0) - i\,r\,R\,\psi_0'(0) - \phi_1^{(3)}(0) + m\,\psi_1^{(3)}(0)$$

```
bc6O2 = Coefficient[bc64, α, 2]
```

— *General::spell1 : Possible spelling error: new symbol name "bc6O2" is similar to existing symbol "bc62".*

$$\frac{i\,\mathrm{c1}\,F\,R\,\phi_0[0]}{(\mathrm{c0}-1)^2} - i\,\mathrm{a1}\,R\,\phi_1[0] - \frac{i\,F\,R\,\phi_1[0]}{\mathrm{c0}-1} + i\,\mathrm{a2}\,r\,R\,\psi_1[0] - i\,\mathrm{c1}\,R\,\phi_0'(0) + 3\,\phi_0'(0) - i\,\mathrm{c0}\,R\,\phi_1'(0) + i\,R\,\phi_1'(0) - 3\,m\,\psi_0'(0) + i\,\mathrm{c1}\,r\,R\,\psi_0'(0) + i\,\mathrm{c0}\,r\,R\,\psi_1'(0) - i\,r\,R\,\psi_1'(0) - \phi_2^{(3)}(0) + m\,\psi_2^{(3)}(0)$$

Here is no tang velocity at bottom wall, bc7

```
bc7 = ψ'[y]
```

$\psi'(y)$

```
bc71 = bc7 /. {ϕ^(a1_)[y] :→ ∂_{y,a1} ϕexp, ψ^(a1_)[y] :→ ∂_{y,a1} ψexp,
    c → cexp, ϕ[y] → ϕexp, U1''[y] → 2 b1, U0 → 1}
```

— *General::spell1 : Possible spelling error: new symbol name "bc71" is similar to existing symbol "bc1".*

$\psi_2'(y)\,\alpha^2 + \psi_1'(y)\,\alpha + \psi_0'(y)$

```
bc72 = bc71 /. y → -n
```

— *General::spell1 : Possible spelling error: new symbol name "bc72" is similar to existing symbol "bc2".*

$\psi_2'(-n)\,\alpha^2 + \psi_1'(-n)\,\alpha + \psi_0'(-n)$

```
bc7O0 = Coefficient[bc72, α, 0]
```

$\psi_0'(-n)$

```
bc7O1 = Coefficient[bc72, α, 1]
```

— *General::spell1 : Possible spelling error: new symbol name "bc7O1" is similar to existing symbol "bc71".*

$\psi_1'(-n)$

```
bc7O2 = Coefficient[bc72, α, 2]
```

— *General::spell1 : Possible spelling error: new symbol name "bc7O2" is similar to existing symbol "bc72".*

$\psi_2'(-n)$

Here is no norm velocity at bottom wall, bc8

```
bc8 = ψ[y]
```

$\psi(y)$

```
bc81 = bc8 /. {ϕ^(a1_)[y] :> ∂_{y,a1} ϕexp, ψ^(a1_)[y] :> ∂_{y,a1} ψexp,
    c → cexp, ϕ[y] → ϕexp, ψ[y] → ψexp, U1''[y] → 2 b1, U0 → 1}
```

— *General::spell1 : Possible spelling error: new symbol name "bc81" is similar to existing symbol "bc1".*

$\psi_2[y]\,\alpha^2 + \psi_1[y]\,\alpha + \psi_0[y]$

```
bc82 = bc81 /. y → -n
```

— *General::spell1 : Possible spelling error: new symbol name "bc82" is similar to existing symbol "bc2".*

$\psi_2[-n]\,\alpha^2 + \psi_1[-n]\,\alpha + \psi_0[-n]$

```
bc8O0 = Coefficient[bc82, α, 0]
```

$\psi_0[-n]$

```
bc8O1 = Coefficient[bc82, α, 1]
```

— *General::spell1 : Possible spelling error: new symbol name "bc8O1" is similar to existing symbol "bc81".*

$\psi_1[-n]$

```
bc8O2 = Coefficient[bc82, α, 2]
```

— *General::spell1 : Possible spelling error: new symbol name "bc8O2" is similar to existing symbol "bc82".*

$\psi_2[-n]$

Here is the solution to the 0 order problem

Here are the equations for the 0 order problem

os1O0

$\phi_0^{(4)}(y)$

os2O0

$\psi_0^{(4)}(y)$

bc1O0

$\phi_0'(1)$

bc2O0

$\phi_0[1]$

bc3O0

$$\frac{\text{a2}\ \phi_0[0]}{1-\text{c0}}+\frac{\text{a1}\ \phi_0[0]}{\text{c0}-1}+\phi_0'(0)-\psi_0'(0)$$

bc4O0

$\psi_0[0]-\phi_0[0]$

bc5O0

$\phi_0''(0)-m\,\psi_0''(0)$

bc6O0

$m\,\psi_0^{(3)}(0)-\phi_0^{(3)}(0)$

bc7O0

$\psi_0'(-n)$

bc8O0

$\psi_0[-n]$

- **Here we solve for the eigen functions**

We solve as much as we can with one command. We leave out the boundary condition with the eigen value in it and treat it later.

```
zerofuncs = DSolve[{os1O0 == 0, os2O0 == 0,
  bc1O0 == 0, bc2O0 == 0, bc4O0 == 0,
  bc5O0 == 0, bc6O0 == 0, bc7O0 == 0, bc8O0 == 0},
 {ϕ0[y], ψ0[y] }, y]
```

$$\left\{\left\{\phi_0[y]\to\frac{1}{6}\left(m\,c_8\,y^3-\frac{2\,m\,(n^3+m)\,c_8\,y^2}{m-n^2}+\frac{m\,((4\,n+3)\,n^2+m)\,c_8\,y}{m-n^2}-\frac{2\,m\,n^2\,(n+1)\,c_8}{m-n^2}\right),\right.\right.$$
$$\left.\left.\psi_0[y]\to\frac{1}{6}\left(c_8\,y^3-\frac{2\,(n^3+m)\,c_8\,y^2}{m-n^2}+\frac{n\,(n^3+m\,(3\,n+4))\,c_8\,y}{n^2-m}-\frac{2\,m\,n^2\,(n+1)\,c_8}{m-n^2}\right)\right\}\right\}$$

Let's choose c_8 so as to make this simple as it is the arbitrary constant for this problem. This makes the two stream functions unity at the interface.

```
zerofuncs1 =
 Simplify[zerofuncs /. C[8] -> -6 (m - n^2) / 2 / m / n^2 / (n + 1)]
```

$$\left\{\left\{\phi_0[y]\to\frac{(y-1)^2\,(2\,n^3+(y+2)\,n^2-m\,y)}{2\,n^2\,(n+1)},\ \psi_0[y]\to\frac{(n+y)^2\,(y\,n^2+m\,(2\,n-y+2))}{2\,m\,n^2\,(n+1)}\right\}\right\}$$

- **Here is a check with eq 8 of Y&H**

```
phi0 = ϕ0[y] /. zerofuncs1[[1]]
```

$$\frac{(y-1)^2\,(2\,n^3+(y+2)\,n^2-m\,y)}{2\,n^2\,(n+1)}$$

```
Simplify[Coefficient[Expand[phi0], y, 3]]
```

$$\frac{n^2-m}{2\,n^2\,(n+1)}$$

```
Simplify[Coefficient[Expand[phi0], y, 2]]
```

$$\frac{n^3+m}{n^2\,(n+1)}$$

```
Simplify[Coefficient[Expand[phi0], y, 1]]
```

$$-\frac{(4\,n+3)\,n^2+m}{2\,n^2\,(n+1)}$$

```
Simplify[Coefficient[Expand[phi0], y, 0]]
```

1

```
psi0 = ψ0[y] /. zerofuncs1[[1]]
```

— *General::spell1 : Possible spelling error: new symbol name "psi0" is similar to existing symbol "phi0".*

$$\frac{(n+y)^2\,(y\,n^2+m\,(2\,n-y+2))}{2\,m\,n^2\,(n+1)}$$

```
Simplify[Coefficient[Expand[psi0], y, 3]]
```

$$\frac{n^2-m}{2\,m\,n^2\,(n+1)}$$

```
Simplify[Coefficient[Expand[psi0], y, 2]]
```

$$\frac{n^3+m}{m\,n^2\,(n+1)}$$

```
Simplify[Coefficient[Expand[psi0], y, 1]]
```

$$\frac{n^3+3\,m\,n+4\,m}{2\,m\,n^2+2\,m\,n}$$

```
Simplify[Coefficient[Expand[psi0], y, 0]]
```

1

- **Here we get the 0 order eigenvalue**

```
bc3O0
```

$$\frac{\mathrm{a2}\,\phi_0[0]}{1-\mathrm{c0}}+\frac{\mathrm{a1}\,\phi_0[0]}{\mathrm{c0}-1}+\phi_0'(0)-\psi_0'(0)$$

```
eig01 =
 bc3O0 //. {ϕ0[0] -> (phi0 /. y -> 0) , ϕ0'[0] -> (D[phi0, y] /. y -> 0),
            ψ0[0] -> (psi0 /. y -> 0) , ψ0'[0] -> (D[psi0, y] /. y -> 0),
   a1 -> (m - n^2) / (n^2 + n),
     a2 -> a1 / m, b1 -> -(m + n) / (n^2 + n)}
```

$$-\frac{2\,n+2}{n\,(n+1)}-\frac{n^2-m}{2\,m\,(n+1)}+\frac{n^2-m}{2\,n^2\,(n+1)}-\frac{2\,n^3+2\,n^2}{n^2\,(n+1)}+\frac{(m-n^2)\,(2\,n^3+2\,n^2)}{2\,(\mathrm{c0}-1)\,n^2\,(n+1)\,(n^2+n)}+\frac{(m-n^2)\,(2\,n^3+2\,n^2)}{2\,(1-\mathrm{c0})\,m\,n^2\,(n+1)\,(n^2+n)}$$

```
eig02 = eig01 /. c0 -> cx + 1
```

$$-\frac{2n+2}{n(n+1)}-\frac{n^2-m}{2m(n+1)}+\frac{n^2-m}{2n^2(n+1)}-\frac{2n^3+2n^2}{n^2(n+1)}+\frac{(m-n^2)(2n^3+2n^2)}{2\,\mathrm{cx}\,n^2(n+1)(n^2+n)}-\frac{(m-n^2)(2n^3+2n^2)}{2\,\mathrm{cx}\,m\,n^2(n+1)(n^2+n)}$$

```
eig03 = Solve[eig02 == 0, cx]
```

$$\left\{\left\{\mathrm{cx}\to-\frac{2n(-m^2+n^2m+m-n^2)}{n^4+4mn^3+6mn^2+4mn+m^2}\right\}\right\}$$

```
c0eig = 1 + FullSimplify[cx /. eig03[[1]]]
```

$$\frac{2(m-1)n(m-n^2)}{n^4+2m(n(2n+3)+2)n+m^2}+1$$

Here we can check with Y&H eq. (9)

```
c00 = 1 + (2 (m - n^2) (m - 1) (n^3 + n^2)) /
  ((n^2 + n) (n^4 + 4 n^3 m + 6 n^2 m + 4 n m + m^2))
```

$$\frac{2(m-1)(m-n^2)(n^3+n^2)}{(n^2+n)(n^4+4mn^3+6mn^2+4mn+m^2)}+1$$

It works!!

```
check = Simplify[c00 - c0eig]
```

— *General::spell1 : Possible spelling error: new symbol name "check" is similar to existing symbol "Check".*

0

■ Here is the solution to the first order problem

■ Here are the equations for the first order problem

```
os1O1
```

$$-i(-2\,\mathrm{b1}\,R\,\phi_0[y]+R(\mathrm{b1}\,y^2+\mathrm{a1}\,y-\mathrm{c0}+1)\phi_0''(y)+i\,\phi_1^{(4)}(y))$$

```
os2O1
```

$$-\frac{i(-2\,\mathrm{b2}\,r\,R\,\psi_0[y]+r\,R(\mathrm{b2}\,y^2+\mathrm{a2}\,y-\mathrm{c0}+1)\psi_0''(y)+i\,m\,\psi_1^{(4)}(y))}{m}$$

bc101

$\phi_1'(1)$

bc201

$\phi_1[1]$

bc301

$$-\frac{\text{a1}\,(\text{c1}\,\phi_0[0]-(\text{c0}-1)\,\phi_1[0])}{(\text{c0}-1)^2}+\frac{\text{a2}\,(\text{c1}\,\phi_0[0]-(\text{c0}-1)\,\phi_1[0])}{(\text{c0}-1)^2}+\phi_1'(0)-\psi_1'(0)$$

bc401

$\psi_1[0]-\phi_1[0]$

bc501

$\phi_1''(0)-m\,\psi_1''(0)$

bc601

$$-i\,\text{a1}\,R\,\phi_0[0]-\frac{i\,F\,R\,\phi_0[0]}{\text{c0}-1}+i\,\text{a2}\,r\,R\,\psi_0[0]-i\,\text{c0}\,R\,\phi_0'(0)+i\,R\,\phi_0'(0)+i\,\text{c0}\,r\,R\,\psi_0'(0)-i\,r\,R\,\psi_0'(0)-\phi_1^{(3)}(0)+m\,\psi_1^{(3)}(0)$$

bc701

$\psi_1'(-n)$

bc801

$\psi_1[-n]$

- **The solution to the first order problem**

We are fortunate again that c1 appears only in bc3. This condition will be saved for last.
We will solve the two OS equations and boundary conditions in three sets because they are so big.

Start with the upper phase OS eq, + top wall BC's

```
eqs12 = Simplify[{os1O1 == 0, bc1O1 == 0, bc2O1 == 0} /.
   {ϕ0[0] -> (phi0 /. y -> 0) ,  ϕ0'[0] -> (D[phi0, y] /. y -> 0),
   ϕ0''[0] -> (D[phi0, {y, 2}] /. y -> 0),
   ϕ0'''[0] -> (D[phi0, {y, 3}] /. y -> 0),
   ϕ0[y] -> phi0, D[ ϕ0[y], {y, a1_}] :> D[phi0, {y, a1}],
   ψ0[0] -> (psi0 /. y -> 0) , ψ0'[0] -> (D[psi0, y] /. y -> 0),
   ψ0''[0] -> (D[psi0, {y, 2}] /. y -> 0),
   ψ0'''[0] -> (D[psi0, {y, 3}] /. y -> 0),
   ψ0[y] -> psi0 , D[ ψ0[y], {y, a2_}] :> D[psi0, {y, a2}],
   c0 -> c0eig }]
```

$$\left\{-i\left(-\frac{\text{b1}\,R\,(2\,n^3+(y+2)\,n^2-m\,y)\,(y-1)^2}{n^2\,(n+1)}+\frac{R\left(\text{b1}\,y^2+\text{a1}\,y-\frac{2\,(m-1)\,n\,(m-n^2)}{n^4+2\,m\,(n\,(2\,n+3)+2)\,n+m^2}\right)((2\,n+3\,y)\,n^2+m\,(2-3\,y))}{n^2\,(n+1)}+i\,\phi_1^{(4)}(y)\right)==0,\ \phi_1'(1)==0,\ \phi_1[1]==0\right\}$$

```
ans12phi = DSolve[eqs12, ϕ1[y], y];
```

— *Solve::svars : Equations may not give solutions for all "solve" variables.*

```
phi1temp1 = ans12phi /. {C[3] -> c3f, C[4] -> c4f};
```

Here is the lower phase OS equation + bottom wall BC's

```
eqs12a = Simplify[{os2O1 == 0, bc7O1 == 0, bc8O1 == 0} /.
   {ϕ0[0] -> (phi0 /. y -> 0) ,  ϕ0'[0] -> (D[phi0, y] /. y -> 0),
   ϕ0''[0] -> (D[phi0, {y, 2}] /. y -> 0),
   ϕ0'''[0] -> (D[phi0, {y, 3}] /. y -> 0),
   ϕ0[y] -> phi0, D[ ϕ0[y], {y, a1_}] :> D[phi0, {y, a1}],
   ψ0[0] -> (psi0 /. y -> 0) , ψ0'[0] -> (D[psi0, y] /. y -> 0),
   ψ0''[0] -> (D[psi0, {y, 2}] /. y -> 0),
   ψ0'''[0] -> (D[psi0, {y, 3}] /. y -> 0),
   ψ0[y] -> psi0 , D[ ψ0[y], {y, a2_}] :> D[psi0, {y, a2}],
   c0 -> c0eig }]
```

— *General::spell1 : Possible spelling error: new symbol name "eqs12a" is similar to existing symbol "eqs12".*

$$\left\{-\frac{1}{m^2}\left(i\left(i\,\psi_1^{(4)}(y)\,m^2-\frac{\text{b2}\,r\,R\,(n+y)^2\,(y\,n^2+m\,(2\,n-y+2))}{n^2\,(n+1)}+\frac{r\,R\left(\text{b2}\,y^2+\text{a2}\,y-\frac{2\,(m-1)\,n\,(m-n^2)}{n^4+2\,m\,(n\,(2\,n+3)+2)\,n+m^2}\right)((2\,n+3\,y)\,n^2+m\,(2-3\,y))}{n^2\,(n+1)}\right)\right)==0,\ \psi_1'(-n)==0,\ \psi_1[-n]==0\right\}$$

```
ans12psi = DSolve[eqs12a, ψ1[y], y];
```

— *General::spell1 : Possible spelling error: new symbol name "ans12psi" is similar to existing symbol "ans12phi".*

— *Solve::svars : Equations may not give solutions for all "solve" variables.*

```
psi1temp2 = ψ1[y] /. ans12psi[[1]]
```

Here are the remaining boundary conditions, except for the one that has c1 in it

```
eqsO1a = ({bc4O1 == 0, bc5O1 == 0, bc6O1 == 0}) /.
  {ϕ0[0] -> (phi0 /. y -> 0) ,  ϕ0'[0] -> (D[phi0, y] /. y -> 0),
   ϕ0''[0] -> (D[phi0, {y, 2}] /. y -> 0),
   ϕ0'''[0] -> (D[phi0, {y, 3}] /. y -> 0),
   ϕ0[y] -> phi0, D[ ϕ0[y], {y, a1_}] :> D[phi0, {y, a1}],
   ψ0[0] -> (psi0 /. y -> 0) , ψ0'[0] -> (D[psi0, y] /. y -> 0),
   ψ0''[0] -> (D[psi0, {y, 2}] /. y -> 0),
   ψ0'''[0] -> (D[psi0, {y, 3}] /. y -> 0),
   ψ0[y] -> psi0 , D[ ψ0[y], {y, a2_}] :> D[psi0, {y, a2}], ϕ1[0] ->
    (phi1temp2 /. y -> 0) ,  ϕ1'[0] -> (D[phi1temp2 , y] /. y -> 0),
   ϕ1''[0] -> (D[phi1temp2 , {y, 2}] /. y -> 0),
   ϕ1'''[0] -> (D[phi1temp2 , {y, 3}] /. y -> 0),
   ϕ1[y] -> phi1temp2 , D[ϕ1[y], {y, a1_}] :> D[phi1temp2 , {y, a1}],
   ψ1[0] -> (psi1temp2 /. y -> 0) ,
   ψ1'[0] -> (D[psi1temp2 , y] /. y -> 0),
   ψ1''[0] -> (D[psi1temp2 , {y, 2}] /. y -> 0),
   ψ1'''[0] -> (D[psi1temp2 , {y, 3}] /. y -> 0),
   ψ1[y] -> psi1temp2 , D[ψ1[y], {y, a1_}] :> D[psi1temp2 , {y, a1}],
   c0 -> c0eig}
```

```
O1bcstemp = Solve[eqsO1a, {C[4], c3f, c4f}]
```

Finally we work on bc3 which contains c1

```
O1bc3 =
 bc3O1 /. {ϕ0[0] -> (phi0 /. y -> 0) ,  ϕ0'[0] -> (D[phi0, y] /. y -> 0),
   ϕ0''[0] -> (D[phi0, {y, 2}] /. y -> 0),
   ϕ0'''[0] -> (D[phi0, {y, 3}] /. y -> 0),
   ϕ0[y] -> phi0, D[ ϕ0[y], {y, a1_}] :> D[phi0, {y, a1}],
   ψ0[0] -> (psi0 /. y -> 0) , ψ0'[0] -> (D[psi0, y] /. y -> 0),
   ψ0''[0] -> (D[psi0, {y, 2}] /. y -> 0),
   ψ0'''[0] -> (D[psi0, {y, 3}] /. y -> 0),
   ψ0[y] -> psi0 , D[ ψ0[y], {y, a2_}] :> D[psi0, {y, a2}], ϕ1[0] ->
    (phi1temp2 /. y -> 0) ,  ϕ1'[0] -> (D[phi1temp2 , y] /. y -> 0),
   ϕ1''[0] -> (D[phi1temp2 , {y, 2}] /. y -> 0),
   ϕ1'''[0] -> (D[phi1temp2 , {y, 3}] /. y -> 0),
   ϕ1[y] -> phi1temp2 , D[ϕ1[y], {y, a1_}] :> D[phi1temp2 , {y, a1}],
   ψ1[0] -> (psi1temp2 /. y -> 0) ,
   ψ1'[0] -> (D[psi1temp2 , y] /. y -> 0),
   ψ1''[0] -> (D[psi1temp2 , {y, 2}] /. y -> 0),
   ψ1'''[0] -> (D[psi1temp2 , {y, 3}] /. y -> 0),
   ψ1[y] -> psi1temp2 , D[ψ1[y], {y, a1_}] :> D[psi1temp2 , {y, a1}],
   c0 -> c0eig}
```

```
bc3O1t2 = Solve[O1bc3 == 0, c1]
```

```
bc3O1t3 = bc3O1t2 /. O1bcstemp[[1]];
```

Here is the answer for c1

```
c1eig1 = bc3O1t3 //. {a1 -> (m - n^2) / (n^2 + n),
    a2 -> a1 / m, b1 -> -(m + n) / (n^2 + n),
    b2 -> b1 / m}
```

Finally if we get the expression by itself and simplify it, the remaining constant, c_3 drops out.

```
see1 = Simplify[c1 /. c1eig1[[1]]]
```

$$\Big(i\,((n+6)\,r\,n^{15} - m\,(n^4+8\,n^3+10\,n^2+126\,n+88)\,r\,n^{12} + m^2\,(2\,r\,n^6+(42\,r-140\,F)\,n^5+(-280\,F+248\,r+224)\,n^4 + (-140\,F-125\,r+1043)\,n^3+(60\,r+1486)\,n^2+32\,(16\,r+21)\,n+224\,r)\,n^9 - m^3\,(32\,r\,n^6+2\,(630\,F+61\,r+112)\,n^5 + (4200\,F-197\,r+651)\,n^4+4\,(1435\,F+156\,r-94)\,n^3 + (3920\,F-356\,r-1306)\,n^2+(1120\,F-2180\,r+406)\,n-1088\,r+944)\,n^8 - m^4\,(8\,(420\,F-2\,r+49)\,n^7+6\,(2520\,F-94\,r+273)\,n^6+2\,(15540\,F+578\,r+429)\,n^5 + 8\,(4620\,F+110\,r-83)\,n^4+(26600\,F-1287\,r+1431)\,n^3 + (11200\,F+258\,r+1524)\,n^2+32\,(70\,F+35\,r-9)\,n+224\,r)\,n^5 - m^5\,(224\,(10\,F+1)\,n^7+32\,(350\,F-9\,r+35)\,n^6+(26600\,F+1524\,r+258)\,n^5 + 3\,(12320\,F+477\,r-429)\,n^4+8\,(3885\,F-83\,r+110)\,n^3 + 2\,(7560\,F+429\,r+578)\,n^2+6\,(560\,F+273\,r-94)\,n+392\,r-16)\,n^4 - m^6\,(16\,(70\,F+59\,r-68)\,n^6+(3920\,F+406\,r-2180)\,n^5+2\,(2870\,F-653\,r-178)\,n^4 + 8\,(525\,F-47\,r+78)\,n^3+(1260\,F+651\,r-197)\,n^2+2\,(112\,r+61)\,n+32)\,n^2 + m^7\,(224\,n^6+32\,(21\,r+16)\,n^5+(-140\,F+1486\,r+60)\,n^4 + (-280\,F+1043\,r-125)\,n^3-4\,(35\,F-56\,r-62)\,n^2+42\,n+2)\,n + m^9\,(6\,n+1) - m^8\,(88\,n^4+126\,n^3+10\,n^2+8\,n+1))\,R\Big) \Big/ \Big(420\,m^2\,(n+1)^2\,(n^4+2\,m\,(2\,n^2+3\,n+2)\,n+m^2)^3\Big)$$

- **Here is a check against the answers of Yiantsios and Higgins**

We need can see that the real part of c1 is 0.

```
Simplify[N[see1 /. {m -> 2, n -> .5, r -> 1, R -> 1, F -> 0}]]
```

$0.00199456\ i$

```
Simplify[N[see1 /. {m -> .4, n -> 4, r -> 2, R -> 10, F -> 5}]]
```

$-10.4682\ i$

Let's try to get figure 2. To produce the ordinate we choose R=1 and take Im[ ] to get rid of the i.

```
plot1 = Plot[Im[N[see1 /. {n -> .5, r -> 1,
      R -> 1, F -> 0}]], {m, 0, 10},
  PlotStyle -> {Dashing[{.03, .02}]}]
plot2 = Plot[Im[N[see1 /. {n -> 1, r -> 1, R -> 1, F -> 0}]], {m, 0, 10}]
```

```
plot3 = Plot[Im[N[see1 /. {n -> 2, r -> 1, R -> 1, F -> 0}]],
  {m, 0, 10}, PlotStyle -> {Dashing[{.04, .02, .02, .02}]}]
```

We seem to have perfect agreement. with fig 2a.

```
Show[{plot1, plot2, plot3}, PlotRange -> {-.01, .02}]
```



- Graphics -

If we try to do figure 2b there is a problem. Y&H state in the text that the flow is neutrally stable if m=√n. Fortunately they tell us why. Looking at equation 2 we see that if m = n^2 then a_1 = a_2 = 0!. If this is the case then by equation 5c, we need c_1 = 0 so that $\phi' - \psi'$ can be finite at y=0.

Thus there is an error in figure 2b. The curve should be m = n^2, not what is plotted!! You can check this further if you need to.